





# A Maxima Guide for Calculus Students

©January 2, 2004 by

Moses Glasner

e-mail: `glasner_AT_math_DOT_psu_DOT_edu`

Address: Department of Mathematics

PSU

University Park, PA 16802

January 2, 2004

<sup>1</sup>See Additional Docs at the Maxima URL: <http://maxima.sourceforge.net/docs.shtml>  
Also see: [http://www.math.psu.edu/glasner/Max\\_doc](http://www.math.psu.edu/glasner/Max_doc)



# 1

## Scientific Calculator

---

By default Maxima comes with its own graphical user interface, called Xmaxima. Those people who are familiar with Emacs may be much more comfortable in using Maxima under the Emacs maxima-mode. This combined with the auctex-mode provides for a very powerful environment for combining computations together with the L<sup>A</sup>T<sub>E</sub>X composition of a document. For very brief usage not requiring extensive interaction, Maxima can be run from a console, such as Xterm under Linux, or DOS prompt under Windows. A more advanced GUI for automating L<sup>A</sup>T<sub>E</sub>X document production, called TeXmacs, can run Maxima and gives a very polished appearance to the output.

We assume that you have loaded Maxima onto your computer and can run it by clicking on a Maxima shortcut or Start menu item. After starting maxima you can scroll through the Primer that appears on the bottom half of the window in which Maxima runs. Also any text appearing there can be copied and pasted as a Maxima instruction. The entire Maxima manual is available in this buffer by clicking on help. Maxima also gives online help by click on Maxima help in the Help menu. The font can be adjusted somewhat by selecting Fonts in the Options menu. Maxima also provides help from the command line in response to the commands describe(something) or example(something). Maxima may ask additional questions before responding to these help requests and the response should be ended with a semicolon ; and then pressing the ENTER key.

Maxima can be used in an interactive manner and a batch processing manner. We will only discuss the interactive manner here using the Xmaxima GUI. Once started instructions for Maxima are entered in the top buffer. Each instruction must end with a semicolon or a dollar sign followed by a press on the Enter key. A semicolon is used if Maxima is to print the output to the screen and dollar sign if the result of the calculation is to be stored internally and used in a subsequent step but not to be displayed on the screen.

When Maxima is ready to process an instruction it displays a C followed by an integer on the screen, for example, (C1). This will be the label assigned to the next instruction entered. Any entered instruction may be retrieved by just entering this label at any time up to the point the Maxima session is stopped or restarted. Maxima assigns a label to the output created by each instruction as well; this begins with a followed by an integer for example (D1). This label is assigned whether or not the display of the output is suppressed by a dollar sign. Or, one can assign a label of one's own choosing to any output Maxima produces by putting "my\_label : " or at the beginning of the input which produces the output. For example *my\_24* : (2 + 3) \* 4 ; or *my\_pie* : 4 \* atan(1);. The labels "my\_24" and "my\_pie " can then be used at any time to retrieve the value assigned to it.

It is important to remember that Maxima does not distinguish between upper case and lower case letters. Thus *Sin(%Pi)* and *sIN(%pI)* produce identical results. However, user defined labels **are case sensitive**. Thus Maxima treats *My\_24* and *my\_24* as different labels.

Since Maxima automatically assigns labels *Cn* and *Dn* to input and output, it is **absolutely essential** not to use these or *cn* and *dn* for any purpose other than recalling a specific input or output. (Since these labels are Maxima assigned, there is no case distinction on these labels.)

For example, the instruction

```
(C1) f(x) := c23x + c24;
```

```
(D1) f(x) := c23x + c24
```

guarantees utter confusion after entering 23 inputs are given in a maxima sessions. Also note that the instruction

```
(C9) kill(all);
```

```
(D0) DONE
```

destroys all information, including loaded packages, entered in a Maxima session. If one needs to use the symbol *f* for some other purpose after entering instruction (C8) but not destroy all information, then the correct instruction would be

```
(C1) kill(all);
```

```
(D0) DONE
```

A very simple use of Maxima is an infinite precision scientific calculator. To

illustrate this, consider the following:

(C1) `fpprec: 60;`

(D1) `60`

(C2) `q : sqrt(2);`

(D2) `SQRT(2)`

(C3) `1 + q;`

(D3) `SQRT(2) + 1`

(C4) `bfloat(q);`

(D4) `1.41421356237309504880168872420969807856967187537694807317668B0`

(C5) `bfloat(1+q);`

(D5) `2.41421356237309504880168872420969807856967187537694807317668B0`

The Maxima variable *fpprec* controls the number of digits Maxima prints to the screen and the command *bfloat(x)* asks it to print out a decimal approximation to the infinite precision value stored in *x*.

From the above you also see that in order to assign an expression to a label or variable one needs to type the label, a colon, and then the expression to be assigned.





# 2

## Finding Limits with Maxima

---

Maxima can be very helpful in checking limits of functions. But first one needs to know how to define functions for Maxima. It is done with a slight addition; specifically, the command is a colon followed immediately by an equal sign, as you see in the following dialogue:

```
(C6) f(x) := sin(x)/x;
```

```
(D6)          SIN(x)
          f(x) := -----
                   x
```

```
(C7) [f(1), f(2), f(1 + h), f(x + h)];
```

```
(D7)          SIN(2)  SIN(h + 1)  SIN(x + h)
          [SIN(1), -----, -----, -----]
                   2          h + 1          x + h
```

and finding limits is easy for Maxima:

```
(C8) limit(f(x), x, 0);
```

```
(D8)          1
```

```
(C9) limit((sin(x+h)-sin(x))/h,h,0);
```

```
((MTEXT) Is ((%SIN SIMP) $x) positive, negative, or zero?)
```

```
positive;
```

```
((MTEXT) Is ((%COS SIMP) $x) positive, negative, or zero?)
```

```
positive;
```

```
(D9)                                COS(x)
```

Note that in the course of the above dialogue Maxima required responses to two questions. After the response, positive, negative, or zero, one needs to type a semicolon and press the ENTER key.

Maxima can even deal with functions defined by several formulas:

```
(C10) f(x) := if x > 0 then 1 else -1;
```

```
(D10)                                f(x) := IF x > 0 THEN 1 ELSE - 1
```

```
(C11) [f(-%pi/3), f(%pi/3)];
```

```
(D11)                                [- 1, 1]
```

Maxima can also be used as a graphing calculator. Try the following instruction:

```
(C12) plot2d(sin(x), [x, 0, 4*%pi]);
```

```
Points = 626
```

```
(D12)                                0
```

# 3

## Limit of Difference Quotient

---

Finding the limit of Difference Quotients is a calculation where Maxima can really be helpful in locating errors. For example, here is a Maxima session finding the limit of the difference quotient of  $\sqrt{x-1}$ .

```
(C1) f(x) := sqrt(x - 1);
(D1)          f(x) := Sqrt(x - 1)
(C2) algebraic : true;
(D2)          TRUE
(C3) diff_quot : (f(x+h) - f(x))/h;
              Sqrt(x + h - 1) - Sqrt(x - 1)
(D3)          -----
              h
(C4) simplified_diff_quot : 1/ratsimp(1/diff_quot);
              1
(D4)          -----
              Sqrt(x + h - 1) + Sqrt(x - 1)
(C5) limit(simplified_diff_quot,h,0);
              Sqrt(x - 1)
(D5)          -----
              2 x - 2
(C6) diff(f(x),x);
```

```

                                1
(D6) -----
                                2 Sqrt(x - 1)

```

In the above dialogue you see several new commands. The last instruction labeled (C6) tells Maxima to just compute the derivative of  $f(x)$  as rapidly as it knows how. It is included only to confirm that the output labeled (D5) is correct.

The command *ratsimp* tells maxima to simplify. However, Maxima by default is set to simplify rational functions and has been told to ignore algebraic (radical) simplifications in order to optimize the other types of simplifications it attempts. However, when we would like it to also perform algebraic simplifications we add the instruction *algebraic : true;*.

If you frequently work with algebraic expressions you may wish to set up an initialization file which maxima reads everytime it (re)starts. To do this you create a textfile called maxima-init.mac consisting of the line *algebraic : true;* and of course any other instructions you would like Maxima to execute at startup. If you are using a word processor, instead of a text editor, for creating this file, then you must save the file as text only. So why not use an editor, like GNU Emacs, instead. You may place this file in your Maxima work directory if you have one or any other directory you choose. However, Maxima must be told to start in this directory. If you are using Windows, then this can be accomplished setting the “Start in” parameter under the Properties of the Maxima icon. (You gain access to its Properties by right-clicking on it.) If Maxima is used under Linux, then the environment variable HOME should be set to the path of directory containing the file maxima-ini.mac.

Maxima can plot two or more graphs simultaneously. The only modification required to the plot2d instruction is enclosing the funtions to be plotting in square brackets. Thus if you would like to see the two graphs in Figure 5 on page 138 plotted on the same axes, then give Maxima the following instruction:

```

(C7) plot2d([f(x),diff(f(x),x)], [x,1.1,2]);
Points = 162
Points = 346
(D7)                                0

```

# 4

## Differentiation Formulas

---

You may be wondering whether you can get Maxima to show you all the differentiation rules without giving it specific functions to differentiate. The answer is in the following dialogue:

```
(C1) depends(f,x);
(D1)                                     [f(x)]
(C2) depends(g,x);
(D2)                                     [g(x)]
(C3) derivative_of_quotient:diff(f/g,x);
                                     df    dg
                                     --  f --
                                     dx    dx
(D3)                                     -- - ----
                                     g      2
                                     g
(C4) ratsimp(derivative_of_quotient);
                                     dg    df
                                     f -- - -- g
                                     dx    dx
(D4)                                     - ----
                                     2
```

g

You see Maxima has the quotient rule memorized in a somewhat different form than the one which is optimal for humans. Also note that the command *depends(f,x)* tells maxima that *f* is a function of *x*. It can then deal with its derivative in a “symbolic” way. If we had not done this, then instruction *diff(f,x)* would have evoked the response 0 from Maxima because it would have thought that *f* and *x* are simply two independent variables.

A variety of instructions control the final form of the answer which Maxima returns. If we wish the answer to be over a common denominator then the instruction is *factor*.

(C5) `f(x) := x^(1/2);`

(D5) 
$$f(x) := x^{1/2}$$

(C6) `g(x) := 1 + x^2;`

(D6) 
$$g(x) := 1 + x^2$$

(C7) `answer:diff(f(x)/g(x),x);`

(D7) 
$$\frac{1}{2 \sqrt{x} (x^2 + 1)} - \frac{3/2 x}{(x^2 + 1)^2}$$

(C8) `factor(answer);`

(D8) 
$$-\frac{3x^2 - 1}{2 \sqrt{x} (x^2 + 1)^2}$$

If you need to access just parts of an answer provided by Maxima, then you can request that Maxima give you labels with which you can access the individual parts. The instruction for this is *pickapart*. For example, if you wish to find where the derivative in Example 10 is equal to 0, then you need a label with which to access the numerator. This is achieved with the following. Note that the additional variable 2 tells Maxima to what “depth” to the expression should be broken into parts. Obviously, you want to use the smallest value that gets the desired label.

```
(C9) pickapart(factor(answer),2);
```

```

          2
(E9) 3x-1

```

```

          2
(E10) 2*SQRT(x)(x+1)

```

```

          E9
(D10) -
          E10

```

You see that Maxima has attached labels (E9) and (E10) to the numerator and denominator of the factored form of the answer. You may wish to see what happens if the 2 is replaced by 1, 3, etc. Now in order to find the zeros of the numerator you do as follows:

```
(C11) solve(E9,x);
```

```
(D11) [x = - 1/SQRT(3), x = 1/SQRT(3)]
```

Note that the answer (D11) is a list with two elements. (Actually everything in Maxima is a list because it is written in the computer language LISP.) You can access the elements which are in the form of equations  $x = \text{something}$  with the instruction **first** and then the value of this solution can be retrieved with the instruction *rhs*

(C12) first(d11);

(D12) 
$$x = - \frac{1}{\text{SQRT}(3)}$$

(C13) value\_of\_x:rhs(D12);

(D13) 
$$- \frac{1}{\text{SQRT}(3)}$$



# 5

## Derivatives of Trig Functions

---

Maxima can be quite helpful in differentiating trig functions. However, a couple of commands specific to trig functions are required in order to instruct Maxima to apply trig identities in simplifications. For example consider the following dialogue:

(C1) `diff( sin(x)/(1 + cos(x)),x);`

$$(D1) \quad \frac{\sin^2(x)}{(\cos(x) + 1)^2} + \frac{\cos(x)}{\cos(x) + 1}$$

(C2) `factor(%);`

$$(D2) \quad \frac{\sin^2(x) + \cos^2(x) + \cos(x)}{(\cos(x) + 1)^2}$$

(C3) `trigsimp(%);`

$$(D3) \quad \frac{\sin^2(x) + \cos^2(x) + \cos(x)}{\cos(x) + 1}$$

2

$$\cos^2(x) + 2 \cos(x) + 1$$

You see that the instruction *trigsimp* was required to get Maxima to make the obvious simplification using the Pythagorean identity. Also note that Maxima does not recognize the shortcut of leaving out parentheses when using trig functions, eg,  $\sin x$ ,  $\cos x$ ,  $\tan x$ ; neither does it recognize the common abuse of notation that puts an exponent directly onto the trig function; i.e.,  $\sin^2(x)$ ,  $\tan^{-1}(x)$  are not recognized; the correct inputs are  $\sin(x)^2$ ,  $\tan(x)^{-1}$ . Also observe that  $\tan(x)^{-1}$  means the reciprocal of tangent and not arctangent which is denoted by  $\text{atan}(x)$ . The other Maxima instruction is *trigreduce* which allows using the multiple angle formulas to reduce powers, eg:

(C4) `factor(cos(x)^2 + 2*sin(x)^2);`

$$2 \quad 2$$

(D4)  $2 \sin^2(x) + \cos^2(x)$

(C5) `trigsimp(cos(x)^2 + 2*sin(x)^2);`

$$2 \quad 2$$

(D5)  $2 \sin^2(x) + \cos^2(x)$

(C6) `trigreduce(cos(x)^2 + 2*sin(x)^2);`

$$\cos(2x) + 1 \quad 1 \quad \cos(2x)$$

(D6)  $\frac{\cos(2x) + 1}{2} + 2 \left( -\frac{1}{2} - \frac{\cos(2x)}{2} \right)$

$$2 \quad 2 \quad 2$$

Of course, one could argue that the latter expression is not simpler than the one we started with. However, it is invaluable for integration, the inverse process of differentiation.

# 6

## The Chain Rule

---

You get Maxima to show how it applies the chain rule by telling it which variables are dependent on which.

```
(C1) f(x) := x^3;
                                     3
(D1)          f(x) := x
(C2) depends(x,u)$
(C3) diff(f(x),u);
                                     2 dx
(D3)          3 x  --
                                     du
```

In the above dialogue we used the functional notation to define  $f$  and used the instruction *depends* to inform Maxima that  $x$  is a function of  $u$  but did not provide a specific formula for it. However, we can also specify the dependence of  $x$  on  $u$  as follows:

```
(C4) kill([x,u]);
(D4)          DONE
(C5) x: sin(u);
(D5)          SIN(u)
(C6) diff(f(x),u);
                                     2
```

(D6)  $3 \cos(u) \sin(u)$

Or, we can use functional notation for both functions and get Maxima to differentiate their composition.

(C7) `kill(x);`

(D7) `DONE`

(C8) `g(x) := sin(x);`

(D8) `g(x) := SIN(x)`

(C9) `diff(f(g(u)),u);`

2

(D9)  $3 \cos(u) \sin(u)$

Note that the instruction *kill* was needed to remove the relationships set in each of the previous dialogues with Maxima. If a variable has multiple dependencies and only one of them is to be removed, then the instruction *remove([u,x],dependency)* can be used.

# 7

## Implicit Differentiation; Higher Derivatives

---

Maxima can easily compute derivatives of implicit functions. Consider the following dialog that instructs Maxima to find  $dy/dx$  given the equation  $x^2 + y^2 = 25$ .

```
(C1) eqn: x^2 + y^2 = 25;

(D1)          2    2
            y  + x  = 25

(C2) depends(y,x);

(D2)          [y(x)]

(C3) deriv_of_eqn:diff(eqn,x);

(D3)          dy
            2 y -- + 2 x = 0
            dx

(C4) solve(deriv_of_eqn,'diff(y,x));

(D4)          dy    x
            [-- = - -]
            dx    y
```

Note the new symbol appearing in the *solve* instruction above. Normally the first argument of *solve* is an equation or list of equations and the second is a variable and a list of variables. Here we see  $dy/dx$  as the second argument. On the other hand there is also a single quote in front. A single quote in front of a symbol tells Maxima to attempt to evaluate the symbol but to treat it as an unknown quantity.

For example,

```
(C5) a : 3;
(D5)                                     3
(C6) b : 4;
(D6)                                     4
(C7) a + b;
(D7)                                     7
(C8) 'a + b;
(D8)                                     a + 4
(C9) 'a + 'b;
(D9)                                     b + a
(C10) a + 'b;
(D10)                                    b + 3
```

So we see that the instruction `solve(deriv_of_eqn, 'diff(y,x))` tells Maxima not try to evaluate the derivative of  $y$  with respect to  $x$  directly (which it really cannot do) but to regard `diff(y,x)` as an unknown quantity and solve for it from the differentiated equation, `deriv_of_eqn`.

The Maxima instruction to find higher order derivatives is the same as that for finding the first derivative except for a third argument indicating the order:

```
(C11) diff(x^n,x,2);
(D11)                                     n - 2
      (n - 1) n x
```

# 8

## Related Rates

---

Maxima can be helpful in solving related rates problems. For example consider the problem of finding the rate of change of the area of a circle given that the rate of change of the radius is  $dr/dt = 60$  when  $t = 2$  and  $r = 120$ :

```
(C1) area: a = %pi*r^2;
                                     2
(D1)                                a = %PI r
(C2) depends([a,r],t)$
(C3) deriv_of_area: diff(area,t);
                                     da      dr
(D3)                                -- = 2 %PI r --
                                     dt      dt
(C4) subst([r=120,diff(r,t)=60],deriv_of_area);
                                     da
(D4)                                -- = 14400 %PI
                                     dt
```

Note that the expression `%PI` is the Maxima's label for  $\pi$ . You might want to see how many digits of  $\pi$  Maxima can find by giving the instruction `fpprec= 1000, bfloat(%PI)`; Also note that Maxima must be told which variables are time-dependent with the instruction `depends`. The instruction `subst` tells maxima to substitute a list of equations appearing as the first argument of `subst`, enclosed in square brackets,

into the expression appearing as its second argument.



# 9

## Linear Approximations and Differentials

---

Maxima has a very powerful built-in tool for finding linear approximations to functions called the Taylor expansion of order 1.

```
(C1) L: Taylor(sin(x),x,%pi/3,1);

                                     %PI
                                     x - ---
                                     SQRT(3)    3
(D1)/T/  ----- + ----- + . . .
                                     2          2

(C2) error1 :float(subst([x=1.1],L)) - float(subst([x=1.1],sin(x)));
(D2)                                     .0012192681247044

(C3) error2 :float(subst([x=1.4],L)) - float(subst([x=1.4],sin(x)));
(D3)                                     .0569768981976797

(C4) plot2d([L,sin(x)], [x,%pi/3-.5,%pi/3+.5]);
```

In the dialogue above the instruction (C1) tells maxima to approximate  $\sin(x)$  by a Taylor polynomial of degree 1, at the point  $x = \pi/3$ . This is simply another name for a linear approximation. On the other hand, Taylor polynomials come in all degrees, and you will have an opportunity to work with them in Math 141 or have a look at the Laboratory Project on page 212. The last argument in the Maxima instruction *Taylor* specifies the degree. By comparing *error1* and *error2* in the above dialogue you can verify that indeed the linear approximation is much more

accurate close to the point of tangency than far away. You can also see this from the graph of the function and its linear approximation produced by Maxima.

You may be wondering about the extra characters */T/* in the output line (D1). This is only Maxima's reminder that the formula it has produced has been truncated and hence cannot be used when great accuracy is needed. If you are really irritated by the choice Maxima makes for its automatic labelling of input and output you can enter something like following instructions:

*inchar : MY\_INPUT*

*outchar : MAXIMA\_OUT*

Or, you can put the following line into your maxima-init.mac file. `tt(inchar : IN_,outchar : OUT_);`

# 10

## Maximum and Minimum Values

---

Maxima can be a great help in checking the process of locating critical points of functions as can see from the following dialogue:

```
(C1) f(x):=x^(3/5)*(4-x);

(D1)          3/5
          f(x) := x  (4 - x)

(C2) deriv_of_f: diff(f(x),x);

(D2)          3 (4 - x)  3/5
          ----- - x
          2/5
          5 x

(C3) pickapart(factor(deriv_of_f),2);

(E3)          4 (2 x - 3)
          2/5

(E4)          5 x
          E3

(D4)          - -
          E4

(C4) solve([E3=0],x);

(D4)          3
          [x = -]
```

Remember that the instruction *pickapart* labels each of the expressions in order that you can refer to it in subsequent instructions. If you are not interested in seeing the intermediate steps, then the instruction *solve(diff(f(x),x);* would also give you the answer that  $x = 3/2$  is one a critical number. However, you would still need to supply the information that the derivative fails to exist at  $x = 0$ .

# 11

## Using Maxima's Lambda definition

---

Maxima can be helpful in seeing that indeed the  $c$  of the Mean Value Theorem depends on the given functions. For example consider the following dialogue which calculates the  $c$  for the functions  $x^2$  and  $x^3$  on the interval  $[0, 1]$

```
(C7) f : lambda([x])$
(C8) MVT(f,a,b):=block([deriv, aver, c], aver:(f(b) - f(a))/(b-a),
                        deriv : diff(f(x),x),
                        c : solve(deriv=aver,x),
                        print(c))$

(C9) g : lambda([x],x^2)$
(C10) MVT(g,0,1)$
      1
[x = -]
      2

(C11) h : lambda([x],x^3)$
(C12) MVT(h,0,1)$
      1          1
[x = - ----, x = ----]
      Sqrt(3)    Sqrt(3)
```

So you see that for  $g$  Maxima found that  $c$  equals  $1/2$  whereas for  $h$  Maxima found different solution with  $1/\sqrt{3}$  being in the interval  $[0, 1]$ .

You also see several new useful instructions. First the declaration involving *lambda* tells maxima that  $g$  is a function of  $x$  given by the formula  $x^2$ . How is this different from the declaration  $g(x) := x^2$  which we have used before? The new declaration requests that Maxima recognize the symbol  $g$  by itself in the definition and/or invocation of another function, where as with the old declaration maxima would not have recognized anything other than  $g(x)$ .

The other new instructions are *block* and *print* in the definition of a function. The first instructs Maxima to make the variables listed in square brackets local to the function being defined by the block and the *print* statement asks Maxima to do just that with all the current values of the variable or quoted strings that are passed as arguments to it.

# 12

## Maxima helps sketch graphs

---

Of course Maxima's plotting ability helps verify graphs which are sketched using calculus techniques. But Maxima can also help implement these techniques. For example, Maxima can find on which intervals a derivative (and second derivative) is positive and on which it is negative. For example, suppose that we are to sketch the graph of the function  $\frac{1}{3}x^3 - 4x = 4$ . Then we find that its derivative is  $x^2 - 4$ , which is zero at  $x = 2$  and  $x = -2$ . We can instruct Maxima to evaluate the derivative at  $x = -3$ ,  $x = 0$  and  $x = 3$  to find where the derivative is positive and where it is negative. We may try this:

```
(C1) expr : x^2- 4$
(C2) subst(-3, x,expr);
(D2)                                     5
(C3) subst(0, x,expr);
(D3)                                     - 4
(C4) subst(3, x,expr);
(D4)                                     5
```

If however, you would want to substitute an algebraic expression for  $x$ , eg,  $u = x - 2$  then Maxima would issue an error message. To inform Maxima that it is supposed to execute the algebra involved in the substitution, the command *ratsubst* needs to be used:

```
(C5) ratsubst(u, x - 2,expr);
2
```

(D5)

$$u + 4 u$$



# 13

## Limits at Infinity and One-sided Limits

---

Maxima is very helpful in checking limits at infinity. Obviously, it can check the extensive algebra that may need to be performed. But in addition it can evaluate the limits directly.

```
(C1) f: (3*x^2 - x - 2)/(5*x^2 + 4*x + 1);
```

```
(D1)          2
          3 x  - x - 2
          -----
          2
          5 x  + 4 x + 1
```

```
(C2) limit(f,x,inf);
```

```
(D2)          3
          -
          5
```

You see that Maxima recognizes the concept of infinity via the label *inf*. On the other hand Maxima does not compute one-sided limits directly. For this purpose the substitution  $x = 1/t$  is very useful as you can see from the following dialogue computing  $\lim_{x \rightarrow 0^-} \frac{|x|}{x}$ .

```
(C3) f : abs(x)/x$
```

```
(C4) limit(ratsubst(1/t,x,f),t,-inf);
```

```
(D4)          - 1
```



# 14

## Plotting In spite of Vertical Asymptotes

---

Using Maxima to check your sketches of graphs of functions requires a little bit of care. For example, if you simply instruct Maxima to plot the function  $f(x) = \frac{x^2}{x^2 - 1}$  by issuing the following instruction you get a nearly useless plot.

```
(C1) plot2d(x^2/(x^2-1), [x, -4, 4])$
```

It does show you the vertical asymptotes; so it is not a complete waste. The problem Maxima encounters is a lack of information about what you want to do with the function values that are too large or too negative to fit on any computer screen. You can instruct Maxima to discard values greater than 10 or less than -10 as indicated in the following dialogue:

```
(C2) plot2d(max(min(x^2/(x^2-1), 10), -10), [x, -4, 4])$
```

Of course the nearly vertical lines are to be interpreted as vertical asymptotes and not portions of the graph.

From this illustration you see that even the most advanced technology cannot replace a human being with good mathematical skills. Although Maxima can help check your execution of the eight steps in sketching the graph of a function, it cannot execute them without your input; it is especially important that you executed Step A and form some understanding of the range of the function and the interval on which you wish to view the graph. However, Maxima can be enormously helpful in checking the accuracy of your work in steps B through H.



# 15

## Optimization; Computing Derivatives of Functions Subject to Constraint

---

Let us illustrate the use of Maxima to solve the following. A window is in the form of a rectangle surmounted by a semicircle. The rectangle is of clear glass while the semicircle is of tinted glass that transmits only half as much light per unit area as clear glass does. The total perimeter is fixed. Find the proportions of the window that will admit the most light. Neglect the thickness of the frame.

Let  $r$  be the radius of the semicircle. Then  $2r$  is the base of the rectangle. Denote the height of the rectangle by  $h$ . Let  $P$  be the total perimeter of the window. Then  $P$  is a fixed constant. We have the following equations connecting  $r$  and  $h$  and expressing the amount of light  $L$  in terms of  $r$  and  $h$ , assuming that  $L$  is one for each unit of area in the rectangular part and  $L$  is  $1/2$  for each unit of area in the semicircular part of the window:

$$P = 2r + 2h + \pi r \qquad L = 2rh + \frac{\pi r^2}{4}$$

(C1) Eqn:  $P = 2*r + 2*h + \%pi*r$

(C2)  $L : 2*r*h + \%pi*r^2/4$

(C3) Soln\_h : solve(Eqn,h);

$$(D3) \qquad [h = - \frac{(\%PI + 2) r - P}{2}]$$

(C4) h\_is : rhs(first(Soln\_h));

$$(\%PI + 2) r - P$$

```

(D4)          - -----
                2

(C5) L_fcn_r : ratsubst(h_is,h,L);

                2
          (3 %PI + 8) r  - 4 P r
(D5)          - -----
                4

(C6) deriv_L :diff(L_fcn_r,r);

          2 (3 %PI + 8) r - 4 P
(D6)          - -----
                4

(C7) Soln_r:solve(deriv_L = 0,r);

          2 P
(D7)          [r = -----]
                3 %PI + 8

```

Since this is the only critical point found for  $L$  on the interval  $[0, P(\pi+2)^{-1}]$  this must give the absolute maximum for  $L$ . Maxima can now be instructed to find the ratio of  $r$  to  $h$ :

```

(C8) r_is : rhs(first(Soln_r));

          2 P
(D8)          -----
          3 %PI + 8

(C9) h_is_now : ratsubst(r_is,r,h_is);

          (%PI + 4) P
(D9)          -----
          6 %PI + 16

(C10) Ratio_finally :ratsimp(r_is/h_is_now);

          4

```

$$(D10) \quad \text{-----}$$

$$\%PI + 4$$

A somewhat different approach to solving problems in this section is to differentiate the constraint equation. This is significantly simpler than solving the constraint equation for one of the variables in case where the constraint equation is complicated or perhaps not even solveable for either of the variables. So in the above problem one would differentiate the equation for the perimeter with respect to  $r$ . This would yield a linear equation in the unknown derivative  $\frac{dh}{dr}$  which can easily be solved. Then one would differentiate the formula for  $L$  with respect to  $r$  and this would also involve the unknown derivative  $\frac{dh}{dr}$  which now can be eliminated. And then one could determine the relationship between  $r$  and  $h$  that yields a critical point. The following dialogue indicates how Maxima can be instructed to carry out this computation.

```
(C1) Eqn: P = 2*r + 2*h + %pi*r$
(C2) depends(h,r);
(D2)                                     [h(r)]
(C3) L : 2*r*h + %pi*r^2/4$
(C4) Deriv_eqn: diff(Eqn,r);

                                     dh
(D4)                                     0 = 2 -- + %PI + 2
                                     dr

(C5) Deriv_h:rhs(first(solve(Deriv_eqn,'diff(h,r)))));

                                     %PI + 2
(D5)                                     - -----
                                     2

(C6) Deriv_L : diff(L,r);

                                     dh      %PI r
(D6)                                     2 -- r + ----- + 2 h
                                     dr        2

(C7) Deriv_L_is : ratsubst(Deriv_h,'diff(h,r),Deriv_L);

                                     (%PI + 4) r - 4 h
```

```
(D7)          - - - - -  
                2  
(C8) solve(Deriv_L_is=0,r);  
                4 h  
(D8)          [r = -----]  
                %PI + 4
```

Note that `'diff(h,r)` must be used instead of `'diff(h,r,r)` in the above dialogue because Maxima must be instructed to treat it as an unknown quantity.



# 16

## Newton's Method; Loading Library Programs; Recursive Functions; Initial Conditions

---

Newton's method can be implemented very easily in Maxima. Actually, there is a Maxima library function which has to be loaded and then all one has to do is give an initial guess. For example the following finds the root of  $x^2 - 2$  closest to 1.5:

```
(C7) load(NEWTON)$
```

```
(C8) f: x^2 - 2$
```

```
(C9) NEWTON(f,1.5);
```

```
Warning: Float to bigfloat conversion of 1.5
```

```
(D9) 1.41421356237469B0
```

The command *NEWTON* is not built into Maxima and must be loaded from the library. Of course, if you wish you can automatically load it each time it is started by placing *load(NEWTON)*; in your maxima-init.mac file. Observe that although Maxima is not case sensitive when it comes to its own commands, it is case sensitive with user labeled and *NEWTON* is regraded as one of the latter. Note that *NEWTON* converts Maxima's output to bigfloat which can be given arbitrary precision. Thus setting *fpprec :100*; would give you square root of 2 to 100 significant digits. Also, the warning about converting float to bigfloat is innocuous. However, it can be avoided by passing 1.5b0 instead of 1.5 to *NEWTON*.

It may be good practice to define your own Maxima function which implements Newton's method.

This definition of *my\_newt* is recursive; i.e. the definition of the function calls the function itself until the desire precision is attained. From the viewpoint of program

clarity this is the method of choice. When calling *my\_newt* lambda definition of function must be used here because the name of a function *f* is being passed to *my\_newt*.

Maxima is very helpful in checking your work on checking antiderivatives that are subject to an initial condition. For example, let us find the antiderivative *v* of  $a = 6t + 4$  subject to the initial condition  $v(0) = 9$

```
(C6) kill(all);
(D0)                                     DONE
(C1) a: 6*t+4;
(D1)                                     6 t + 4
(C2) v: integrate(a,t);
(D2)                                     2
                                     3 t  + 4 t
(C3) c_val:solve(subst(0,t,v+c)=-6,c);
(D3)                                     [c = - 6]
(C4) v: v+rhs(first(c_val));
(D4)                                     2
                                     3 t  + 4 t - 6
(C5) s: integrate(v,t);
(D5)                                     3      2
                                     t  + 2 t  - 6 t
(C6) c_val:solve(subst(0,t,s+c)=9,c);
(D6)                                     [c = 9]
(C7) s+rhs(first(c_val));
(D7)                                     3      2
                                     t  + 2 t  - 6 t + 9
```

You see that Maxima chronically omits the constant of integration when it performs the instruction *integrate*; so it is up to the user to provide it.

# 17

## Limits of Riemann sums; The Definite Integral

---

The actual evaluation of the limit of the Riemann sums associated to a function on an interval  $[a, b]$  is absolutely tedious. So we are very pleased that Maxima can help. Maxima's instruction for summing a series is *sum* and it takes four arguments: the expression, the index, the beginning value of the index, and the ending value of the index. Maxima's default is not to carry out the evaluation of the sum of a series. And the instruction to change this is: *simpsum : true*\$. The Maxima dialogue for doing is as follows:

```
(C1) R_sum: (3/n)*sum((2+3*i/n)^4,i,1,n);
```

$$\begin{array}{c} n \\ \text{====} \\ \sqrt[3]{\frac{(2 + \frac{3i}{n})^4}{n}} \\ \text{====} \\ i = 1 \end{array}$$

```
(D1)
```

$$\frac{\text{-----}}{n}$$

```
(C2) simpsum:true$
```

```
(C3) R_sum: ratsimp((3/n)*sum((2+3*i/n)^4,i,1,n));
```

$$6186 n^4 + 9135 n^3 + 3510 n^2 - 81$$

```
(D3) -----  
4  
10 n  
(C4) limit(R_sum,n,\inf);  
3093  
(D4) ----  
5  
  
\newpage  
(C5) integrate(x^4,x,2,5);  
3093  
(D5) ----  
5
```

The formulas 5 and 6 given on page 329 do not need to be memorized. However, you do need to know all the others on that page. You also need to know all the properties of integrals listed on pages 333 to 325. If you think about their meanings in terms of simple geometric properties of areas, then they are all extremely simple. The purpose of the figures on these pages is to help you think in this way.

# 18

## Fundamental Theorem of Calculus

---

You may be curious whether Maxima knows the Fundamental Theorem. The following dialogue provides the answer.

```
(C1) S: lambda( [x], integrate(cos(t),t,0,x));
(D1)          LAMBDA([x], INTEGRATE(COS(t), t, 0, x))
(C2) diff(S(x),x);
((MTEXT) Is  $x  positive, negative, or zero?)
positive;
(D2)          COS(x)
(C3) diff(S(x^3),x);
((MTEXT) Is  $x  positive, negative, or zero?)
positive;
(D3)          2      3
              3 x  COS(x )
```

Note that for Maxima's sake it is essential to use a different name for the integration variable than for the variable of the function  $S$ . When Maxima executes these steps it requires knowledge of the sign of  $x$  and it asks for the user's input. The correct response is *positive*;. The semicolon and a press of the enter key are needed to indicate the end of the response.



# 19

## Indefinite Integrals; Using ode2

---

The Maxima instruction *integrate* serves two purposes; which one depends on the number of arguments it the instruction is passed. It is used for telling Maxima to find either the indefinite or definite integral. So, how does Maxima know which to find? By the number of arguments; with two arguments, the expression and its variable its the indefinite integral but if the lower and upper limits are also passed to the instruction, then it is the definite integral that Maxima is being asked to find.

Another instruction that can be used to achieve the same purpose is *ode2*. It can be used to instruct Maxima to solve the equation  $\frac{dy}{dx} = f(x)$ , for  $y$ . This is a very powerful instruction but in this case using a sledge hammer to drive a tack does no harm. The following indicates how this is done.

```
(C3) kill(all)$
```

```
(C1) eq: 'diff(y,x) = sqrt(1/x^4-1/x^5);
```

```
(D1)      dy      1      1
          -- = SQRT(-- - --)
          dx      4      5
                   x      x
```

```
(C2) ode2(eq,y,x);
```

```
(D2)      x - 1 3/2
          2 (-----)
                   x
          y = ----- + %C
```

Whenever, Maxima is handed an expression it automatically tries to evaluate it. So remember that the quote mark before the *diff* is necessary to tell Maxima not to waste time trying to evaluate  $\frac{dy}{dx}$  because that is not the issue Maxima will be asked to work on here. Maxima is being told what this derivative is and the next instruction will ask it to solve for  $y$ .

Note that when you use *ode2* Maxima includes the arbitrary constant  $\%C$  in its response.

If you were to try to solve this problem “by hand”, you would first need to factor  $x^{-2}$  out of the square root and then make the substitution  $u = 1 - x^{-1}$ .



# 20

## Areas between Curves, Finding Intersections

---

Maxima can be very helpful in finding the points of intersection of curves. The following dialogue demonstrates how:

```
(C1) eq1 : y = x - 1$
```

```
(C2) eq2 : y^2 = 2*x + 6$
```

```
(C3) solve([eq1,eq2],[x,y]);
```

```
(D3) [[x = - 1, y = - 2], [x = 5, y = 4]]
```

Note that Maxima does not integrate absolute value. Thus to get Maxima to do a problem where neither graph remains above the other on the entire interval, you first need to find the points of intersection and then instruct Maxima to sum the integrals of the upper function minus the lower function over the intervals between.



# 21

## Volumes by slicing; Using Trigsimp

---

Maxima cannot help with finding the formula for the area of the cross-section of a typical slice  $A(x)$  but it is certainly capable of checking the ensuing integration.

For example, finding the volume generated by revolving the area below  $y = \sec(x)$ , above  $y = 1$  and between  $x = 1$  and  $x = -1$  requires evaluating  $V = \int_{-1}^1 \sec^2(x) - 1^2 dx$

```
(C1) A : (sec(x))^2 - 1$
```

```
(C2) B : trigsimp(A);
```

```
(D2)          2
          COS (x) - 1
          -----
          2
          COS (x)
```

```
(C3) integrate(B,x,-1,1);
```

```
(D3)          2 TAN(1) - 2
```

Of course, Maxima could have found the integral without issuing the instruction *trigsimp*. This instruction causes Maxima to use the Pythagorean identities to simplify an expression. An instruction like *ratsimp* would have no effect here.



# 22

## Volumes by Cylindrical Shells; Using Trigreduce and Trigexpand

---

Let find the volume of the solid generated by revolving the region between  $y = \cos x$ ,  $y = 0$  and the lines  $x = 0$ ,  $x = \pi/2$  about the  $x$ -axis. Using the shell method, the integral to be calculated is

$$2\pi \int_0^1 y \cos^{-1} y dy$$

which is another difficult integral requiring integration by-parts. However, the disk method yields an integral which can easily be found:

$$\pi \int_0^{\pi/2} \cos^2 x dx = \frac{\pi}{2} \int_0^{\pi/2} 1 + \cos 2x dx = \frac{\pi}{2} \left[ x + \frac{1}{2} \sin 2x \right]_0^{\pi/2} = \frac{\pi^2}{4}$$

If you forget some of the double/half angle formulas, then you ask Maxima to remind you;

halfangles: true;

(C1) A : cos(x)^2\$

(C2) B: trigreduce(A);

COS(2 x) + 1

(D2)

-----

2

(C3) C: trigexpand(B);

2            2

- SIN (x) + COS (x) + 1

(D3)

-----

2

From this dialogue you see that *trigreduce* tells Maxima to use the half-angle formulas to simplify an expression and *trigexpand* tells Maxima to use the double-angle formulas. Note that the commonly accepted mathematical abuse of notation of using  $\cos^2(x)$  to stand for  $\cos(x)^2$  is not recognized by Maxima.